Integrais em coordenadas polares

Nesta seção ilustramos como o Maple V pode ser usado para calcular certas integrais usando coordenadas polares: $x = r\cos(\theta)$, $y = r\sin(\theta)$.

Suponha que voce queira integrar a seguinte funcao

$$f(x,y) = \frac{1}{(x^2+y^2)^{\left(\frac{3}{2}\right)}}$$

sobre a região anelar *R* que está no primeiro quadrante limitada entre os dois circulos:

$x^2+y^2 = 1$ e $x^2+y^2 = 4$.

e a s retas

$y = 0$ e $y = x$.

Como primeiro passo vamos plotar a região *R* usindo Maple V e coordenadas retangulares. A regiao R plotada pelos seguintes comandos.

```
> restart;
> with(plots):
> P1 :=plot(sqrt(1-x^2),x=1/sqrt(2)..1):
> P2 :=plot(sqrt(4-x^2),x=sqrt(2)..2):
> P3 := plot(x,x=1/sqrt(2)..sqrt(2)): # The line y = x
> display({P1,P2,P3},view = [0..2,0..2],scaling=constrained);
```



Podemos usar integral iterada para calcular esta integral dupla? Em geral isto é complicado e pode exigir que quebremos a região em varias subregiões. Entretanto, a integral fica completamente simples se usarmos coordenadas polares. Vamos plotar a região R em coordenadas polares.

```
> P1 := plot([2,t,t=0..Pi/4],0..2,0..2,coords=polar): # Circilo r= 2

> P2 := plot([1,t,t=0..Pi/4],0..2,0..2,coords=polar): # Circulo r =1

> P3 := plot([t,Pi/4,t=1..2],0..2,0..2,coords=polar): # reta y = x

> display({P1,P2,P3},scaling=constrained);
```



Em coordenadas polares a integral é a seguinte:

> **I1 := Int(Int((1/r^3)*r,r=1..2),theta=0..Pi/4);**

$$I1 := \int_0^{\frac{1}{4}\pi} \int_1^2 \frac{1}{r^2}\, dr\, d\theta$$

> **I1 := value(I1);**

$$I1 := \frac{1}{8}\pi$$

ou,

> **int(int(1/r^2,r=1..2),theta=0..Pi/4);### int com i pequeno!**

$$\frac{1}{8}\pi$$

É interessante observar que se plotamos a região R usando o comando **plot** mas omitimos **coords = polar** obtemos uma região retangular que dá os limites de integração do problema.

> **P1 := plot([2,t,t=0..Pi/4],0..2,0..2):**

> **P2 := plot([1,t,t=0..Pi/4],0..2,0..2):**

> **P3 := plot([t,Pi/4,t=1..2],0..2,0..2):**

> **display({P1,P2,P3},scaling=constrained);**



A integral de uma função $F(r, \theta)$ sobre esta região tem os mesmos limites como integral sobre a região em coordenadas polares. A regra é que o integrando em coordenadas (u, v) se transforma:

$\mathrm{f}(\mathrm{x}(u, v), \mathrm{y}(u, v))$ | $\mathrm{jacobiano}(\mathrm{x}(u, v), \mathrm{y}(u, v))$ | $du dv$

Por exemplo, no problema o integrando é

$$\frac{1}{(x^2 + y^2)^{\left(\frac{3}{2}\right)}}.$$

Vamoso usar o determinante Jacobiano para determinar o integrando usando coordendas polares e Maple V:

Primeiro precisamos chamar um pacote de álgebra linear e usar os procedimentos **jacobian** e **det** .

> **with(linalg):**

Warning, new definition for norm

Warning, new definition for trace

Agora vamos calcular o valor absoluto do determinate da matriz jacobiana.

```
> dA :=abs((det(jacobian([r*cos(theta),r*sin(theta)],[r,theta]))))*dr*dtheta;
```

$$dA := \left| \cos(\theta)^2 r + r \sin(\theta)^2 \right| dr\, dtheta$$

```
> dA := simplify(dA);
```

$$dA := |r|\, dr\, dtheta$$

```
> F := subs(x=r*cos(theta),y=r*sin(theta),1/(x^2+y^2)^(3/2));
```

$$F := \frac{1}{(r^2 \cos(\theta)^2 + r^2 \sin(\theta)^2)^{3/2}}$$

```
> F := simplify(abs(F));
```

$$F := \frac{1}{|r|^3}$$

```
> simplify(expand(F)*dA,symbolic);
```

$$\frac{dr\, dtheta}{r^2}$$

O ultimo resultado dá o integrando que foi usado neste exemplo.

Exemplo: Converter a integral iterada

$$\int_{-1}^{0} \int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}} f(x, y)\, dy\, dx$$

para uma integral iterada usando coordenadas polares.

Solução: A região *R* de integração é o semi disco a esquerda do eixo y. Veja o desenho.

```
> plot({[1,t,t=Pi/2..3*Pi/2],[t,Pi/2,t=0..1],[t,3*Pi/2,t=0..1]},
```

```
> coords = polar,scaling = constrained,
```

> **xtickmarks=2,ytickmarks=2);**



>

Se omitirmos **coords = polar** obtemos uma região retangular no plano (r, θ) que da os limites de integração.

> **plot({[1,t,t=Pi/2..3*Pi/2],[t,Pi/2,t=0..1],[t,3*Pi/2,t=0..1]},**

> **0..1,0..3*Pi/2,**

> **scaling = constrained,xtickmarks=2,ytickmarks=2);**



>

Agora substituindo dxdy pela expressão do jacobiano completamos a solução do exemplo.

> **Int(Int(subs(x=r*cos(theta),y=r*sin(theta),f(x,y))*det(jacobian(**

> **[r*cos(theta),r*sin(theta)],[r,theta])),r=0..1),theta=Pi/2..3*Pi/2);**

$$\int_{\frac{1}{2}\pi}^{\frac{3}{2}\pi}\int_{0}^{1} \mathrm{f}(r\cos(\theta), r\sin(\theta))\,(\cos(\theta)^2 r + r\sin(\theta)^2)\,dr\,d\theta$$

> **simplify(");**

$$\int_{\frac{1}{2}\pi}^{\frac{3}{2}\pi}\int_{0}^{1} \mathrm{f}(r\cos(\theta), r\sin(\theta))\,r\,dr\,d\theta$$

Exemplo: Vamos considerar o problema de determinar a área dentro do circulo $r = 3\sin(\theta)$ e a direita da reta x = 7/8.

Solução: Primeiro vamos plotar a região usando o comando **polarplot** que é parte do pacote **plots** .

```
> with(plots):
```

O próximo comando plota o circulo.

```
> P1 := polarplot(3*sin(theta),theta=0..Pi):";
```




```
>
```

A reta vertical x = 7/8 tem equação

```
> r = 7/8*sec(theta);
```

$$r = \frac{7}{8}\sec(\theta)$$

```
> P2 := polarplot(7/8*sec(theta),theta=0..5*Pi/12):
```

Precisamos determinar a area da região dentro do circulo e a direita da reta.

```
> display({P1,P2});
```



Para fazer isto precisamos determinar os valores de θ onde a reta e o circulo se encontram.

Digamos se sejam θ_1 e θ_2 , respectivamente, entao a area é dada pela seguinte integral duplal.

$$\int_{\theta_1}^{\theta_2} \int r\,dr = \frac{7\sec(\theta)}{8}\,d\theta$$

Para determinar os pontos de interseção ou olhamos o grafico (se os pontos forem complicados em geral não dá para ver) ou usamos o comando **fsolve** para determinar os pontos de interseção, para isto precisamos informar o intervalo onde queremos procurar a solução.

```
> plot({7/8*sec(theta),3*sin(theta)},theta=0..5*Pi/12);
```



>

Isto sugere que as interseções ocorrem nos intervalos [0.2,0.4], e [1,1.3]. Vamos usar **fsolve** para estimar estes pontos com 10 digitos de precisão.

> **theta[1] := fsolve(7/8*sec(x)=3*sin(x),x,0.2..0.4);**

$$\theta_1 := .3114132927$$

> **theta[2] := fsolve(7/8*sec(x)=3*sin(x),x,1..1.3);**

$$\theta_2 := 1.259383034$$

Finalmente, a seguinte integral dupla dá uma muito boa aproximação da área.

> **Int(Int(r,r=7/8*sec(theta)..3*sin(theta)),theta=theta[1]..theta[2]);**

$$\int_{.3114132927}^{1.259383034} \int_{\frac{7}{8}\sec(\theta)}^{3\sin(\theta)} r \, dr \, d\theta$$

> **value(");**

$$1.066876285$$

Assim a área procurada é aproximadamente 1.066876285.

>

Exercicios

1. Evalie a integral dupla de y^2 sobre a regiao *R* dada no primeiro quadrante, fora do circulo $r = 2$ e dentro da cardioide $r = 2\,(1 + \cos(\theta))$.

2. Encontre o volume V da solido acima do retangulo polar

R = {(r, θ)| $1 \le r$, $r \le 3$, $0 \le \theta, \theta \le \frac{\pi}{4}$ } e sob a superficie $z = x\, \mathbf{e}^{\left(\frac{x^2 + y^2}{\pi}\right)}$.

3. Evalie $\int_0^1 \int_0^1 \sqrt{4 - x^2 - y^2}\, dx\, dy$.